OPAL Technical Note TN607 April 20, 1999

Implementation of the Higgsstrahlung Matrix Element

Jenny Böhme Benno List

RWTH Aachen CERN III. Physikalisches Institut A EP division

Jenny.Bohme@cern.ch Benno.List@cern.ch

Abstract

To exploit the kinematic properties of standard model Higgs production it is useful to introduce the corresponding matrix element for this process as variable into a likelihood selection. This matrix element can be calculated with a small routine which is presented in this note.

1 Introduction

Matrix elements have been used in a number of likelihood selections at OPAL, most notably the matrix element for 4–jet production in Z^0/γ^∗ → qq̄gg events and the four fermion matrix elements from Excalibur [Ber95] for e^+e^− → Z^0Z^0 and e^+e^− → W^+W^− events [OPA99].

These matrix elements can also be used in a search for Higgs bosons to discriminate Higgs prodution events from background events. This is especially important for analyses which are intended for Higgs searches in general two– Higgs–doublet models, which should be independent of the quark flavor the Higgs decays into and therefore cannot make use of b–tagging variables to suppress the two and four fermion background.

However, the use of these matrix elements can be augmented by the use of the matrix element of the process under consideration, that is, Higgsstrahlung.

2 The Higgsstrahlung Matrix Element

The Higgsstrahlung four fermion matrix element has first been calculated in [Ber85]. Since the Higgs is a scalar, the matrix element does not depend on the decay angle of the Higgs, but it depends on the invariant mass and the decay angle of the Z^0 boson.

Equation (5.2) of this paper gives the five–dimensional cross section for the process e^+(p_+)e^−(p_−) → Z^∗ → Z^0 H(q_H) → μ^+(q_+) μ^−(q_−) qq̄, where the symbols in parantheses indicate the corresponding four–vectors, as

d^5σ = α^3 2π^2 g^2 ( M^2_Z s ) |⃗q_H|[C_1(p_+q_−)(p_−q_+) + C_2(p_+q_+)(p_−q_−)] R (s) R (s_1) dq^0_H dΩ_H dΩ_μ.

The symbols have the following meanings:

α Electromagnetic coupling constant R (s) = (s − M^2_Z)^2 + M^2_ZΓ^2_Z Z^0 propagator M_Z Z^0 pole mass Γ_Z Z^0 width s = (p_+ + p_−)^2 = 4 E^2_b center–of–mass energy squared s_1 = 4 E_b (E_b − q^0_H) + m^2_H Invariant mass squared of the muons E_b Beam energy m_H = √ q^2_H Higgs rest mass C_1 = (c^2_v + c^2_a)^2 + 4c^2_vc^2_a C_2 = (c^2_v + c^2_a)^2 − 4c^2_vc^2_a g = (sin θ_W cos θ_W)^{−1} Weak coupling constant θ_W Weinberg angle c_a = −^1_4 g Axial vector coupling for muons c_v = −^1_4 g(4 sin^2 θ_W − 1) Vector coupling for muons In general, c_a and c_v are given by c_a =^1_2 g t_{3,f} Axial vector coupling for fermion f c_v =^1_2 g (t_{3,f} − 2q_f sin^2 θ_W) Vector coupling for fermion f t_{3,f} Third component of weak isospin: +^1_2 for ν_ℓ, u, c, t; −^1_2 for ℓ^−, d, s, b Q_f Electric charge: 0 for ν_ℓ; −1 for ℓ^−; +^2_3 for u, c, t; −^1_3 for d, s, b.

To obtain the matrix element from this cross section expression, one must divide by the phase space density (PSD). The needed three-particle PSD can be obtained recursively from the two–particle PSD [PDG98, p. 186, eq. (35.12)]:

dΦ_3 (Q, q_+, q_−, q_H) = dΦ_2 (q_Z, q_+, q_−) · dΦ_2 (Q, q_Z, q_H) · (2π)^3dq^2_Z,

with q_Z = q_+ + q_− and Q = q_H + q_Z. In our notation, q^2_Z can be expressed as

q^2_Z = (Q − q_H)^2 = ( √ s − q^0_H)^2 − ⃗q^2_H = s − 2 √ s q^0_H + m^2_H = s_1,

therefore we have dq^2_Z = −2 √ s dq^0_H.

The two–particle phase space for the Z^0 decay is given by (neglegting the fermion masses) dΦ_2 (q_Z, q_+, q_−) = 1 2(4π)^2 dΩ_μ,

and the phase space for the Higgs–Z^0 system is

dΦ_2 (Q, q_Z, q_H) = 1 (4π)^2 |⃗q_H| √ s dΩ_H, resulting in dΦ_3 (Q, q_+, q_−, q_H) = 1 32π |⃗q_H| d q^0_H d Ω_H d Ω_μ.

For massles initial particles, the differential cross section for any 2 → n reaction is is given by [PDG98, p. 187, eq. (35.24)]:

dσ =

(2π)^4 s |M|^2 dΦ_n,

therefore the matrix element is

|M|^2 = s (2π)^4 dσ dΦ_n = α^3 π^3 g^2 m^2_Z

C_1(p_+q_−)(p_−q_+) + C_2(p_+q_+)(p_−q_−) R (s) R (s_1) . 3 Calculation of the Matrix Element

The calculation of the matrix element has been implemented in a small routine, called HIGGSM. The routine is called with 5 arguments, three of them inputs:

• IMODE selects the fermion flavor of the particles the Z^0 decays to. 1 = νν̄, 2 = ℓ^−ℓ^+, 3 = uū, cc̄, 4 = dd̄, ss̄, bb̄. This parameter determines the vector and axial vector couplings.

• GAMMAZ is the width (in GeV) of the Z^0 boson. The value of the matrix element varies strongly with the invariant mass of the complete hadronic system (which determines the off-shellness of the intermediate Z^0 boson) and of the Z^0 decay particles. This parameter can be set to a value that is larger than the measured Z^0 width to compensatethe finite experimnetal resolution of the Z^0 mass.

• P is a set of 7 four–vectors:

– P (*, 1): 4–vector of the decay fermion of the Z^0 (i. e., q_−). – P (*, 2): 4–vector of the decay antifermion of the Z^0 (i. e., q_+). – P (*, 3): 4–vector of the first decay particle of the H. – P (*, 4): 4–vector of the second decay particle of the H. – P (*, 5): Sum of the above 4–vectors.

– P (*, 6): 4–vector of the incoming beam electron (i. e., p_−). – P (*, 7): 4–vector of the incoming beam positron (i. e., p_+).

Since s_1 is calculated from the invariant mass of the Z^0 decay particles, P (*, 3) and P (*, 4) are actually not used. These two rows of the array are still reserved to make the call similar to other matrix element calculation routines.

Two variables are returned from the routine:

• AME1 is the square of the Higgsstrahlung matrix element, i. e. |M|^2.

• AME2 is the square of the Higgsstrahlung matrix element, i. e. |M|^2, that results from an interchange of the Z^0 decay fermion and antifermion. In many cases the user will want to average over these or take the maximum, and since it implies almost no additional cost to calculate the second matrix element, it is provided.

4 Application in the Analysis

In principle it is rather easy to use the matrix element information in an analysis, especially if the analysis is based on artificial neural networks or likelihood techniques. After reconstruction of the event, the four–vectors of the final state particles are handed over to HIGGSM, and the returned matrix element is used as an additional variable in the analysis.

However, several details have to be considered:

• Should on use measured four–vectors, or four–vectors returned from a 4C, 5C, or 5^1_2 C kinematic fit?

• Which value for the Z^0 width should one use?

• How should one treat the case where the assignment between measured particles or jets and fermions is not certain?

• For which decay fermion flavour should one calculate the matrix element in the case of Z^0 decays to quarks?

4.1 Choice of Input Four{Vectors and of the Z^0 Width

No definite answer exists to the question which kind of four–vector one should use as input. In the end, one has to try several possibilities and find out which yields the best result. However, some observations may help in the decision:

The matrix element depends most strongly on the two Z^0 propagator terms, which correspond to the intermediate (at LEP2: virtual) and the final (at LEP2: real) Z^0 boson. Using particle momenta that result from a kinematic fit which imposes energy and momentum conservation completely removes this dependence on s^′. In a situation where the amount of energy carried away by initial state radiation is typically smaller than the s^′ resolution, this is probably good.

Likewise, using momenta from a fit that constrains one dijet mass to the Z^0 mass removes or (in the case of a “soft” constraint as in the 5^1_2 C fit [Böh99]) softens the dependency on the reconstructed Z^0 mass. This may be helpful, especially if the reconstructed Z^0 mass is already exploited in other ways, e. g. by using the fit probability of a 5^1_2 C fit.

Especially when unfitted four–vectors are used, the average deviation of the reconstructed Z^0 boson mass from its nominal value is much stronger than expected from the Z^0 width alone, which may lead to a “washing out” of the information from angular distributions, which is also contained in the matrix element. Therefore one should investigate whether a calculation with a value for GAMMAZ that reflects the experimental resolution rather than the natural width of the Z^0 boson improves the result. Remember here that Γ is the full– width half–maximum value of the Breit–Wigner function, not the RMS. The full–width half–maximum value of a Gaussian with RMS σ is √ 8 ln 2 σ ≈ 2.35 σ.

4.2 Assignment of Jets to Fermions

In the Higgs search channels where the Z^0 decays to electrons or muons, the matrix element can be unambigously determined, since the flavour and charge of the Z^0 decay fermions are known. If, on the other hand, the Z^0 and the H decay hadronically, one has the usual jet–pairing problem, plus an additional one due to the fact that the matrix element depends on the quark/antiquark assignment of the jets.

One approach that is often taken is to evaluate the matrix element for all possible jet–quark assignments (there are 12 in the four jet channel, since the quark/antiquark is irrelevant for the Higgs decay jets), and use the maximum value of the matrix element as likelihood variable.

If one does not want to make any assumption on the quark–fermion assignment it seems much more motivated to average the matrix element (or, precisely, its square) over all possible associations. By doing this, one treats the quark flavour as unobserved quantity (which it is), and uses a flavour–averaged matrix element. This is completely analogous to the omnipresent use of spin– averaged matrix elements. In other words: By averaging the matrix element over all possible jet–fermion associations, one symmetrizes it with respect to these assignments, and gets a matrix element that reflects the probability to get a particular jet configuration from a particular process, rather than the most probable fermion configuration. This seems to be preferable on theoretical and aesthetic grounds, but again only the final performance is the ultimate judge.

A second possibility is to use the jet–pairing which has been obtained already in the analysis, e. g. by applying a b–tag. In that case only the two matrix elements have to be considered (and, by the above arguments, should be averaged) that result from exchanging the fermion/antifermion assignment of the two Z^0 decay particles.

4.3 Choice of Quark Flavours

In the case of hadronic Z^0 decays there are two different Z^0H matrix elements, one for the Z^0 decaying to down–type (d, s, b) quarks, on for up–type (u, c) quarks. The Z^0 decays approximately 66 % of the time to down–type quarks, and 34 % of the time to up–type quarks, so one sensible way to proceed is to use a correspondingly weighted average of the matrix elements from up– and down–type quarks.

The second possiblity is to use only the down–type matrix element, in order to better separate Z^0 → bb̄ decays from Higgs decays. Again, both possibilities should be tried out.

5 Results for the Four–Jet Channel

In this section, some distributions are shown that compare a Higgs signal to various background sources. All distributions are made from fully simulated and reconstructed Monte Carlo events with a centre–of–mass energy of 189 GeV.

Figure 1 shows the distribution of the Z^0H matrix element for Higgs masses of 71, 81, and 91 GeV in comparison to the distributions for two and four fermion backgrounds. It is clearly visible that the matrix element has some discrimination power for all these Higgs masses, and that the Z^0H matrix element is typically bigger for Z^0H events than for background events. In general the discrimination power is bigger if one uses momenta from a 5^1_2 C fit than from a 4C fit.

Figure 2 shows some examples how ratios of the Z^0H matrix element to matrix elements from the various background sources can discriminate between different processes. All plots are made for a 91 GeV Higgs signal.

For the W^+W^− background, the four fermion matrix element for W^+W^− production from Excalibur has been used [Tor99]. The matrix element was evaluated using jet momenta from a 4C fit, and was averaged over all possible jet–parton assignments. A good separation power is achieved if one uses either momenta from a 4C or a 5^1_2 C fit to evaluate the Z^0H matrix element.

Even for the background from Z^0Z^0 production, which is especially hard to seperate from the signal of a 91 GeV Higgs, some separation is visible. The Z^0Z^0 matrix element was taken from Excalibur, evaluated using 4C fit momenta, and averaged over all possible jet–parton assignments. It is interesting to note that the ratio |M_{ZH}|/|M_{ZZ}| is, on average, bigger for Z^0H events than for Z^0Z^0 events. Indeed the average value of the Z^0Z^0 matrix element is bigger for Z^0H events. This effect is currently not understood.

Finally, a comparison is made between a Higgs signal and the QCD (i. e. Z^0/γ^∗ → qq̄) background. The QCD matrix element |M_{QCD}| was taken from the Event generator [Cat96]. It was evaluated from the measured jet momenta, and the highest value for all possible jet–parton assignments was taken. Again, a nice separation of signal and background is achieved using the matrix elements.

a) b)

c) d)

e) f)

log jMZHj log jMZHj

log jMZHj log jMZHj

log jMZHj log jMZHj

M_H =71 GeV

M_H =81 GeV

M_H =91 GeV

4C Fit Momenta 5^1_2C Fit Momenta

Figure 1: Comparison of the logarithm of the Higgsstrahlung matrix element for Z^0H events (shaded), four fermion background (\\ hatched), and QCD background (// hatched) for different Higgs masses. In the figures on the left side (a, c, e), the 4C fit momenta were used to evaluate the matrix element. In the right hand figures (b, d, f), the 5^1_2 C fit momenta were used. In all cases the jet pairing was used that resulted in the highest 5^1_2 C fit probability.

a) b)

c) d)

e) f)

log(jM_{ZH}j=jM_{WW}j) log(jM_{ZH}j=jM_{WW}j)

log(jM_{ZH}j=jM_{ZZ}j) log(jM_{ZH}j=jM_{ZZ}j)

log(jM_{ZH}j=jM_{QCD}j) log(jM_{ZH}j=jM_{QCD}j)

W^+W� Background

Z^0Z^0 Background

Z^0/γ� Background

4C / 5^1_2C 5^1_2C / 5^1_2C

4C / 5C 4C / 5^1_2C

4C / 5^1_2C 5^1_2C / 5^1_2C

Figure 2: Separation power of ratios of matrix elements for different kinds of backgrounds. All plots shown here compare a 91 GeV Higgs signal (shaded) to various sources of background (hatched histograms). For each source of background, the distribution of the logarithm of the ratios of the Z^0H matrix element to the corresponding background process matrix element is shown. The Z^0H matrix element is evaluated either using 4C (a, c, d, e) or 5^1_2 C (b, f) fit momenta, and either the jet pairing with the best 5C (c) or 5^1_2 C (a, b, d, e, f) fit probability is used. 8

References

[Ber85] F. A. Berends and R. Kleiss (1985): Initial–state radiation at LEP energies and the corrections to Higgs boson production. — Nucl. Phys. B260, 32–60.

[Ber95] F. A. Berends, R. Pittau and R. Kleiss (1995): Excalibur: A Monte Carlo program to evaluate all four fermion processes at LEP– 200 and beyond. — Comp. Phys. Commun. 85, 437–452.

[Böh99] J. Böhme (1999): Alternative kinematic fits for the Higgs boson search in the 4–jet–channel. — OPAL technical note TN593, 22 pp., unpublished.

URL: http://www.cern.ch/opal/doc/technote/html/tn593.html.

[Cat96] S. Catani and M. H. Seymour (1996): The dipole formalism for the calculation of QCD jet cross–sections at next–to–leading order. — Phys. Lett. B378, 287–301.

[PDG98] C. Caso, Particle Data Group et al. (1998): Review of particle physics. — Eur. Phys. J. C3, 1–794.

[OPA99] OPAL Collaboration (1999): W^+W^− production in e^+e^− collisions at 189 GeV. — OPAL physics note PN378, 16 pp., unpublished. URL: http://www.cern.ch/Opal/pubs/pn/html/pn378.html.

[Tor99] E. Torrence (1999): Use of the Excalibur four–fermion matrix elements in a likelihood selection. — OPAL technical note in preparation.

A Implementation

*.**********************************************************************
      SUBROUTINE HIGGSM (IMODE, GAMMAZ, P, AME1, AME2)
*.
*...HIGGSM Calculate Higgsstrahlung matrix element
*. according to formula (5.2) in
*. F.A. BEHREND and R. KLEISS: Initial state radiation at LEP energies
*. and the corrections to Higgs boson production.
*. Nucl. Phys. B260(1985)32-60.
*. For details, especially normalization of the matrix element,
*. see OPAL-TN607
*.
*. INPUT : IMODE : 1: Z -> nu nubar
*. 2: Z -> l- l+
*. 3: Z -> u ubar
*. 4: Z -> d dbar
*. INPUT : GAMMAZ : "effective" width of Z0
*. INPUT : P : P(I, 1): 4-vector of decay particle of Z0
*. INPUT : P : P(I, 2): 4-vector of decay ANTIparticle of Z0
*. INPUT : P : P(I, 3): 4-vector of 1st quark of h (not used!)
*. INPUT : P : P(I, 4): 4-vector of 2nd quark of h (not used!)
*. INPUT : P : P(I, 5): sum of 4-vectors 1 to 4
*. INPUT : P : P(I, 6): 4-vector of incoming beam electron
*. INPUT : P : P(I, 7): 4-vector of incoming beam positron
*. OUTPUT : AME1 : squared matrix element for Higgsstrahlung
*. OUTPUT : AME2 : squared matrix element for configuration where
*. decay quarks of Z0 are interchanged
*.
*. SEQUENCE :
*. COMMON :
*. CALLS :
*. CALLED : <User>
*.
*. AUTHOR : Benno List, Jenny Boehme
*. VERSION : 1.10
*. CREATED : 5-APR-99
*. LAST MOD : 13-APR-99
*.
*. Modification Log.
*. BL 13-APR-99: Added correct normalization, streamlined the code
*.
*.**********************************************************************

      IMPLICIT NONE

* --- Arguments
      INTEGER IMODE
      REAL GAMMAZ, P (4, 7), AME1, AME2

* --- Variables

*- Some fundamental constants

      REAL SIN2THW, AMASSZ, PI, ALPHA
      PARAMETER (SIN2THW = 0.23124)
      PARAMETER (AMASSZ = 91.187)
      PARAMETER (PI = 3.14159265359)
      PARAMETER (ALPHA = 7.2973581119E-3)

      INTEGER I, IM
      LOGICAL LFIRST

      REAL AMASZ2, GAMMZ2
      REAL CA2, CV2, C1 (4), C2 (4), T3 (4), Q (4), G2, ANORM
      REAL S, S1, RSRS1
      REAL PPLQPL, PPLQMI, PMIQPL, PMIQMI

      REAL R

* --- Data statements
      DATA LFIRST /.TRUE./

*- T3: Weak isospin of nu, e, u, d
      DATA T3 /0.5, -0.5, 0.5, -0.5/
*- Q: Charge of nu, e, u, d
      DATA Q /0., -1., 0.666666667D0, -0.333333333D0/

* --- Statement functions
      R(S) = (S - AMASZ2)**2 + AMASZ2*GAMMZ2

* --- Statements

*- Set up important constants
      IF (LFIRST) THEN
        LFIRST = .FALSE.
        AMASZ2 = AMASSZ**2
        GAMMZ2 = GAMMAZ**2
        G2 = 1. / (SIN2THW*(1-SIN2THW))
        ANORM = (ALPHA/PI)**3 * G2 * AMASZ2

        DO I = 1, 4
          CA2 = 0.25*G2 * T3(I)**2
          CV2 = 0.25*G2 * (T3(I) - 2.*Q(I)*SIN2THW)**2
          C1 (I) = (CV2 + CA2)**2 + 4*CV2*CA2
          C2 (I) = (CV2 + CA2)**2 - 4*CV2*CA2
        END DO
      END IF

*- Check that IMODE is in allowed range
      IF (IMODE .GE. 1 .AND. IMODE .LE. 4) THEN
        IM = IMODE
      ELSE
        PRINT *, ’HIGGSM: Wrong IMODE! Using default (4).’
        IM = 4
      END IF

*- Calculate s
      S = MAX (0., P (4,5)**2 - P (1,5)**2 - P (2,5)**2 - P (3,5)**2)

*- s1 is really the invariant mass squared of the Z decay particles,
*- use that relation, not the one given in the paper
      S1 = (P (4, 1) + P (4, 2))**2 - (P (1, 1) + P (1, 2))**2 -
     - (P (2, 1) + P (2, 2))**2 - (P (3, 1) + P (3, 2))**2

*- for comparison with BEHREND&KLEISS:
*- q_minus = p (*, 1)
*- q_plus = p (*, 2)
*- p_minus = p (*, 6)
*- p_plus = p (*, 7)

      PPLQPL = P(4,7)*P(4,2)-P(1,7)*P(1,2)-P(2,7)*P(2,2)-P(3,7)*P(3,2)
      PPLQMI = P(4,7)*P(4,1)-P(1,7)*P(1,1)-P(2,7)*P(2,1)-P(3,7)*P(3,1)
      PMIQPL = P(4,6)*P(4,2)-P(1,6)*P(1,2)-P(2,6)*P(2,2)-P(3,6)*P(3,2)
      PMIQMI = P(4,6)*P(4,1)-P(1,6)*P(1,1)-P(2,6)*P(2,1)-P(3,6)*P(3,1)
      RSRS1 = R (S)*R (S1)

*- for normalization see technical note xxx

      AME1 = ANORM / RSRS1 *
     * MAX (0., (C1(IM)*PPLQMI*PMIQPL + C2(IM)*PPLQPL*PMIQMI))

      AME2 = ANORM / RSRS1 *
     * MAX (0., (C1(IM)*PPLQPL*PMIQMI + C2(IM)*PPLQMI*PMIQPL))

      RETURN
      END